

SCILAB FOR CHEMICAL ENGINEERS

TSEC - ONLINE CERTIFICATE COURSE

SAMPLE PROBLEMS

ADITYA GANESH

Date: 08 - 02 - 2025

Numerical Methods continued...

1 Numerical Integration

Let's consider the following data points:

x	y
0.0	0.0000
0.5	0.4794
1.0	0.8415
1.5	0.9975
2.0	0.9093
2.5	0.5985
3.0	0.1411
3.5	-0.3508
4.0	-0.7568
4.5	-0.9775
5.0	-0.9589
5.5	-0.7055
6.0	-0.2794

Table 1: Discrete data points for numerical integration

1.1 Trapezoidal Method

Compute the integral of the given data points using the trapezoidal method.

$$\int_a^b f(x) dx \approx \sum_{i=1}^{N-1} \frac{(x_{i+1} - x_i)}{2} [f(x_{i+1}) + f(x_i)] \quad (1)$$

```

1 // Trapezoidal Method for discrete data
2 function I = trapezoidal_data(x, y)
3     n = length(x);
4     I = 0;
5     for k = 1:n-1
6         I = I + 0.5 * (x(k+1) - x(k)) * (y(k) + y(k+1));
7     end
8 endfunction
9
10 // Data points
11 x = [0, 0.5, 1.0, 1.5, 2.0, 2.5, 3.0, 3.5, 4.0, 4.5, 5.0,
12      5.5, 6.0];
13 y = [0, 0.4794, 0.8415, 0.9975, 0.9093, 0.5985, 0.1411,
14      -0.3508, -0.7568, -0.9775, -0.9589, -0.7055, -0.2794];
15
16 I_trap = trapezoidal_data(x, y);
17 disp(I_trap);

```

1.2 Simpson's 1/3rd Rule

Compute the integral of the given data points using Simpson's 1/3rd rule.

$$\int_a^b f(x) dx \approx \frac{h}{3} [f(x_1) + 4f(x_2) + 2f(x_3) + 4f(x_4) + \dots + 2f(x_{n-2}) + 4f(x_{n-1}) + f(x_n)] \quad (2)$$

$$\int_a^b f(x) dx \approx \frac{x_2 - x_1}{3} \left[f(x_1) + f(x_N) + 4 \sum_{i=2}^{N-1} f(x_{\text{even}}) + 2 \sum_{i=3}^{N-1} f(x_{\text{odd}}) \right] \quad (3)$$

```

1 // Simpson's 1/3rd Rule for discrete data
2 function Int = Simpsons_one_third(x,y)
3     Int = (x(2) - x(1))/3 * (y(1) + y($) + 4* sum(y(2:2:$-1))
4         + 2* sum(y(3:2:$-1)));
5 endfunction

```

```

6 // Data points
7 x = [0, 0.5, 1.0, 1.5, 2.0, 2.5, 3.0, 3.5, 4.0, 4.5, 5.0,
      5.5, 6.0];
8 y = [0, 0.4794, 0.8415, 0.9975, 0.9093, 0.5985, 0.1411,
      -0.3508, -0.7568, -0.9775, -0.9589, -0.7055, -0.2794];
9
10 disp(Simpsons_one_third(x,y))

```

1.3 Simpson's 3/8th Rule

Compute the integral of the given data points using Simpson's 3/8th rule.

$$\int_a^b f(x) dx \approx \frac{3h}{8} (f(x_1) + 3f(x_2) + 3f(x_3) + 2f(x_4) + 3f(x_5) + \dots f(x_n)) \quad (4)$$

```

1 // Simpson's 3/8th Rule for discrete data
2 function Int = Simpsons_three_eighth(x,y)
3     Int = 3*(x(2) - x(1))/8 * ( y(1) + y($ ) + 3*sum(y(2:$-1))
4         - sum(y(4:3:$-1)))
5
6 // Data points
7 x = [0, 0.5, 1.0, 1.5, 2.0, 2.5, 3.0, 3.5, 4.0, 4.5, 5.0,
      5.5, 6.0];
8 y = [0, 0.4794, 0.8415, 0.9975, 0.9093, 0.5985, 0.1411,
      -0.3508, -0.7568, -0.9775, -0.9589, -0.7055, -0.2794];
9
10 disp(Simpsons_three_eighth(x,y))

```

2 Regression

2.1 Line Fitting (Linear Regression)

Fit a straight line to the following data points and compute the coefficients of the line $y = ax + b$.

Data points:

$$x = [1, 2, 3, 4, 5]$$

$$y = [2, 3, 5, 4, 6]$$

The linear regression formula for the slope a and intercept b is given by:

$$a = \frac{n \sum xy - \sum x \sum y}{n \sum x^2 - (\sum x)^2}$$

$$b = \frac{\sum y - a \sum x}{n}$$

The normal equations for linear regression in the matrix form are:

$$\mathbf{A}\mathbf{X} = \mathbf{B}$$

Where:

$$\mathbf{A} = \begin{bmatrix} \sum x^2 & \sum x \\ \sum x & n \end{bmatrix}$$

$$\mathbf{X} = \begin{bmatrix} a \\ b \end{bmatrix}$$

$$\mathbf{B} = \begin{bmatrix} \sum xy \\ \sum y \end{bmatrix}$$

```
1 function [a, b] = linear_regression(x, y)
2     n = length(x);
3
4     // Calculate sums
5     sum_x = sum(x);
6     sum_y = sum(y);
7     sum_xy = sum(x .* y);
8     sum_x2 = sum(x .^ 2);
9
10    // Calculate slope (a) and intercept (b)
11    a = (n * sum_xy - sum_x * sum_y) / (n * sum_x2 - sum_x^2)
12        ;
13    b = (sum_y - a * sum_x) / n;
14 endfunction
```

```

15 // Data points
16 x = [1, 2, 3, 4, 5];
17 y = [2, 3, 5, 4, 6];
18
19 // Perform linear regression
20 [a, b] = linear_regression(x, y);
21
22 disp("Coefficients of the line: ");
23 disp("Slope (a): " + string(a));
24 disp("Intercept (b): " + string(b));

```

2.2 Polynomial Fitting (Quadratic Polynomial)

Fit a quadratic polynomial $y = ax^2 + bx + c$ to the following data points and compute the coefficients.

Data points:

$$x = [1, 2, 3, 4, 5]$$

$$y = [2, 3, 5, 4, 6]$$

The normal equations for quadratic polynomial fitting are.

$$\mathbf{A} = \begin{bmatrix} \sum x^4 & \sum x^3 & \sum x^2 \\ \sum x^3 & \sum x^2 & \sum x \\ \sum x^2 & \sum x & n \end{bmatrix}$$

$$\mathbf{X} = \begin{bmatrix} a \\ b \\ c \end{bmatrix}$$

$$\mathbf{B} = \begin{bmatrix} \sum x^2 y \\ \sum xy \\ \sum y \end{bmatrix}$$

Solve $\mathbf{AX} = \mathbf{B}$:

```

1 function [a, b, c] = polynomial_regression(x, y)
2     n = length(x);
3

```

```

4      // Calculate sums
5      sum_x = sum(x);
6      sum_y = sum(y);
7      sum_x2 = sum(x .^ 2);
8      sum_x3 = sum(x .^ 3);
9      sum_x4 = sum(x .^ 4);
10     sum_xy = sum(x .* y);
11     sum_x2y = sum((x .^ 2) .* y);
12
13     // Construct matrices
14     A = [sum_x4, sum_x3, sum_x2;
15          sum_x3, sum_x2, sum_x;
16          sum_x2, sum_x, n];
17     B = [sum_x2y; sum_xy; sum_y];
18
19     // Solve for coefficients
20     X = A \ B;
21
22     a = X(1);
23     b = X(2);
24     c = X(3);
25 endfunction
26
27 // Data points
28 x = [1, 2, 3, 4, 5];
29 y = [2, 3, 5, 4, 6];
30
31 // Perform polynomial regression
32 [a, b, c] = polynomial_regression(x, y);
33
34 disp("Coefficients of the quadratic polynomial: ");
35 disp("a: " + string(a));
36 disp("b: " + string(b));
37 disp("c: " + string(c));

```

3 Ordinary Differential Equations

3.1 Euler's Method

Solve the first-order ODE $\frac{dy}{dt} = -2y$ with initial condition $y(0) = 1$.

Hint:

$$y_{n+1} = y_n + hf(x_n, y_n) \quad (5)$$

```
1 // Euler's Method for solving ODE
2 function [t, y] = euler_method(f, t0, y0, tf, h)
3     t = t0:h:tf;
4     y = zeros(1, length(t));
5     y(1) = y0;
6     for i = 1:length(t)-1
7         y(i+1) = y(i) + h * f(t(i), y(i));
8     end
9 endfunction
10
11 // Define the ODE as a function
12 function dydt = Euler_ode(t, y)
13     dydt = -2 * y;
14 endfunction
15
16 // Initial condition
17 t0 = 0;
18 y0 = 1;
19 tf = 5;
20 h = 0.1;
21
22 // Solve the ODE using Euler's method
23 [t, y] = euler_method(Euler_ode, t0, y0, tf, h);
24
25 // Plot the solution
26 plot(t, y);
27 xlabel("t");
28 ylabel("y(t)");
29 title("Solution of the Simple ODE using Euler's Method");
```

3.2 Runge-Kutta Method

Solve the first-order ODE $\frac{dy}{dt} = y - t^2 + 1$ with initial condition $y(0) = 0.5$.

Hint:

$$y_{i+1} = y_i + \frac{1}{6}(k_1 + 2k_2 + 2k_3 + k_4)$$

$$k_1 = hf(x_i, y_i)$$

$$k_2 = hf\left(x_i + \frac{h}{2}, y_i + \frac{k_1}{2}\right)$$

$$k_3 = hf\left(x_i + \frac{h}{2}, y_i + \frac{k_2}{2}\right)$$

$$k_4 = hf(x_i + h, y_i + k_3)$$

```
1 // Runge-Kutta Method for solving ODE
2 function [t, y] = runge_kutta(f, t0, y0, tf, h)
3     t = t0:h:tf;
4     y = zeros(1, length(t));
5     y(1) = y0;
6     for i = 1:length(t)-1
7         k1 = h * f(t(i), y(i));
8         k2 = h * f(t(i) + h/2, y(i) + k1/2);
9         k3 = h * f(t(i) + h/2, y(i) + k2/2);
10        k4 = h * f(t(i) + h, y(i) + k3);
11        y(i+1) = y(i) + (k1 + 2*k2 + 2*k3 + k4) / 6;
12    end
13 endfunction
14
15 // Define the ODE as a function
16 function dydt = RK_ode(t, y)
17     dydt = y - t^2 + 1;
18 endfunction
19
20 // Initial condition
21 t0 = 0;
22 y0 = 0.5;
```



```

23 tf = 2;
24 h = 0.1;
25
26 // Solve the ODE using Runge-Kutta method
27 [t, y] = runge_kutta(RK_ode, t0, y0, tf, h);
28
29 // Plot the solution
30 plot(t, y);
31 xlabel("t");
32 ylabel("y(t)");
33 title("Solution of the Intermediate ODE using Runge-Kutta
    Method");

```

3.3 2nd order differential equation: Runge-Kutta Method

Solve the second-order ODE $\frac{d^2y}{dt^2} + 3\frac{dy}{dt} + 2y = 0$ with initial conditions $y(0) = 0$ and $\frac{dy}{dt}(0) = 1$.

Hint: Convert the second-order ODE into a system of first-order ODEs:

$$\frac{dy_1}{dt} = y_2$$

$$\frac{dy_2}{dt} = -3y_2 - 2y_1$$

```

1 // Runge-Kutta Method for solving ODE
2 function [t, y] = runge_kutta(f, t0, y0, tf, h)
3     t = t0:h:tf;
4     y = zeros(length(y0), length(t));
5     y(:, 1) = y0;
6     for i = 1:length(t)-1
7         k1 = h * f(t(i), y(:, i));
8         k2 = h * f(t(i) + h/2, y(:, i) + k1/2);
9         k3 = h * f(t(i) + h/2, y(:, i) + k2/2);
10        k4 = h * f(t(i) + h, y(:, i) + k3);
11        y(:, i+1) = y(:, i) + (k1 + 2*k2 + 2*k3 + k4) / 6;
12    end
13 endfunction

```

```

14
15 // Define the system of ODEs as a function
16 function dydt = RK_2_ode(t, y)
17     dydt = [y(2); -3 * y(2) - 2 * y(1)];
18 endfunction
19
20 // Initial conditions
21 t0 = 0;
22 y0 = [0; 1];
23 tf = 5;
24 h = 0.1;
25
26 // Solve the ODE using Runge-Kutta method
27 [t, y] = runge_kutta(RK_2_ode, t0, y0, tf, h);
28
29 // Plot the solution
30 plot(t, y(1, :));
31 xlabel("t");
32 ylabel("y(t)");
33 title("Solution of the Complex ODE using Runge-Kutta Method")
    ;

```